

From The Edge:

by

A. Razzak Memon

Pick and Print PDFs On-Demand

R:BASE 7.0 for Windows

Trademarks:

R:BASE, Pocket R:BASE, Oterro, R:Capture, R:Charts, R:Code, R:DWF Viewer, R:Java, R:Linux, R:Merge, R:Scheduler, R:Scope, R:Struc,, R:Syntax, R:Style and R:Tango are trademarks or registered trademarks of R:BASE Technologies, Inc. All Rights Reserved. All other brand, product names, company names and logos are trademarks or registered trademarks of their respective companies.

January 25, 2004

From the Edge: Pick and Print PDFs On-Demand in R:BASE 7.0 for Windows
Section: Plugins
Chapter: Running R:BASE Your Way!
Platform: R:BASE 7.0 for Windows
Build: 7.0.70.30107 and higher

The Next Generation R:BASE 7.0 for Windows includes the option of designing your own custom Plugins to integrate in your applications.

LoadFileName.RBL Plugin can be used in your R:BASE 7.0 for Windows applications to automate the selection of any specific type of file to load and/or to LAUNCH. There could be various other reasons to make the best of this plugin.

Syntax: **PLUGINS LoadFileName.RBL VarName|<options>**

VarName The variable name to return text results.

Options: FULLPATH (ON/OFF)
 TITLE <value>
 FILTER <value>
 INITIAL_DIR <value>
 MULTISELECT (ON/OFF)
 SHOW_HIDDEN (ON/OFF)
 NO_LONG_NAMES (ON/OFF)
 NO_NETWORK_BUTTON (ON/OFF)
 HIDE_READ_ONLY (ON/OFF)
 NO_DEREFERENCE_LINKS (ON/OFF)
 DONT_ADD_TO_RECENT (ON/OFF)
 ENABLE_SIZING (ON/OFF)
 OLD_STYLE (ON/OFF)
 NO_VALIDATE (ON/OFF)

Each parameter must be separated by pipe "|" symbol.

FULLPATH (ON/OFF) Default: OFF

When used, specifies the resulting variable includes full path of the selected file.

TITLE

Specifies the text in the dialog's title bar.

FILTER

Determines the file masks (filters) available in the dialog. The file-selection dialog includes a drop-down list of file types under the edit box. When the user picks a file type from the list, only files of the selected type are displayed in the dialog.

To configure file masks at design time, click on the ellipsis marks (...) to the right of the Filter property in the Object Inspector. This opens the Filter editor. In the left column of the Filter editor, under Filter Name, type a brief description of each file type that will be available at runtime. In the right column, under Filter, type the file mask corresponding to each description. For example, the description "Adobe PDF Files" might appear to the left of the mask "*.pdf", and the description "Adobe PDF Files" might appear to the left of the mask "*.pdf". Since the description appears in the drop-down list at runtime, it is often helpful to show the mask explicitly in the description (for example, "Adobe PDF Files (*.pdf)").

To create file masks in program code, assign a value to the Filter property that consists of a description and a mask separated by a vertical bar (pipe) character. Do not include spaces around the vertical bar. For example, if you would like to only display the list of Adobe PDF files, use the following:

```
PLUGINS LoadFileName.RBL vFileName +  
|FULLPATH OFF +
```

```
|TITLE Select File +  
|FILTER Adobe PDF Files (*.pdf)#*.PDF
```

Multiple filters should be separated by # symbol. For example, if you would like to display the list of Adobe PDF files as well as R:BASE Command file, use the following:

```
PLUGINS LoadFileName.RBL vFileName +  
|FULLPATH OFF +  
|TITLE Select File +  
|FILTER Adobe PDF Files (*.pdf)#*.PDF#Command Files (*.rmd)#*.RMD
```

To include multiple masks in a single filter, separate the masks with semicolons and make sure to enclose the entire syntax within quotes.

Example:

```
PLUGINS LoadFileName.RBL 'vFileName +  
|FULLPATH OFF +  
|TITLE Select File +  
|FILTER R:BASE Command Files (*.cmd,*.rmd,*.eep)#*.CMD;*.RMD;*.EEP'
```

If no value is assigned to FILTER, the dialog displays all file types.

INITIAL_DIR

Determines the current directory when the dialog opens.

INITIAL_DIR determines the default directory displayed in the file-selection dialog when it opens. For example, to point the dialog at the RBTI\RBG7\Samples directory, set the value of INITIAL_DIR to C:\RBTI\RBG7\Samples. If no value is assigned to INITIAL_DIR, or if the specified directory does not exist, the initial directory will be the current working directory. Otherwise, the dialog opens with either the current working directory or the My Documents directory, depending on the version of Windows.

Example:

```
PLUGINS LoadFileName.RBL vFileName +  
|FULLPATH OFF +  
|TITLE Select File +  
|FILTER Adobe PDF Files (*.pdf)#*.PDF +  
|INITIAL_DIR C:\RBTI\RBG7\Technical Documents
```

MULTISELECT (ON/OFF) Default: OFF

Allows users to select more than one file in the dialog.

Example:

```
PLUGINS LoadFileName.RBL vFileName +  
|FULLPATH OFF +  
|TITLE Select File +  
|FILTER Adobe PDF Files (*.pdf)#*.PDF +  
|INITIAL_DIR C:\RBTI\RBG7\Technical Documents+  
|MULTISELECT ON
```

SHOW_HIDDEN (ON/OFF) Default: OFF
Ensures that hidden files are also visible in the dialog.

Example:

```
CLS
PLUGINS LoadFileName.RBL vFileName +
|FULLPATH OFF +
|TITLE Select File +
|FILTER All Files (*.*)#*. * +
|INITIAL_DIR C:\ +
|MULTISELECT OFF +
|SHOW_HIDDEN ON
```

NO_LONG_NAMES (ON/OFF) Default: OFF
Displays 8.3-character file names only. This option is only valid if parameter option also includes OLD_STYLE ON.

Example:

```
CLS
PLUGINS LoadFileName.RBL vFileName +
|FULLPATH OFF +
|TITLE Select File +
|FILTER All Files (*.*)#*. * +
|INITIAL_DIR C:\RBTI\RBG7\Technical Documents +
|MULTISELECT OFF +
|SHOW_HIDDEN OFF +
|NO_LONG_NAMES ON +
|OLD_STYLE ON
```

NO_NETWORK_BUTTON (ON/OFF) Default: OFF
Removes the Network button (which opens a Map Network Drive dialog) from the file-selection dialog. Applies only if parameter option also includes OLD_STYLE ON.

Example:

```
CLS
PLUGINS LoadFileName.RBL vFileName +
|FULLPATH OFF +
|TITLE Select File +
|FILTER All Files (*.*)#*. * +
|INITIAL_DIR C:\RBTI\RBG7\Technical Documents +
|MULTISELECT OFF +
|SHOW_HIDDEN OFF +
|NO_LONG_NAMES ON +
|OLD_STYLE ON +
|NO_NETWORK_BUTTON ON
```

HIDE_READ_ONLY (ON/OFF) Default: OFF
Removes the Open As Read Only check box from the dialog.

Example:

```
PLUGINS LoadFileName.RBL vFileName +
|FULLPATH OFF +
|TITLE Select File +
|FILTER All Files (*.*)#*. * +
|INITIAL_DIR C:\ +
|MULTISELECT OFF +
|SHOW_HIDDEN OFF +
|NO_LONG_NAMES OFF +
|OLD_STYLE OFF +
|NO_NETWORK_BUTTON OFF +
|HIDE_READ_ONLY ON
```

NO_DEREFERENCE_LINKS (ON/OFF) Default: OFF
Disables dereferencing of Windows shortcuts. If the user selects a shortcut, assigns to File Name the path and file name of the shortcut itself (the .LNK file), rather than the file linked to the shortcut.

DONT_ADD_TO_RECENT (ON/OFF) Default: OFF
Prevents the file from being added to the list of recently opened files.

ENABLE_SIZING (ON/OFF) Default: OFF (Windows 98 and later)
Lets the Explorer-style dialog be resized with the mouse or keyboard. By default, the dialog allows this resizing regardless of the value of this option. It is only required if you provide a hook procedure or custom template. (Old Style dialogs never permit resizing.)

OLD_STYLE (ON/OFF) Default: OFF
Creates the older style of file-selection dialog.

NO_VALIDATE (ON/OFF) Default: OFF
Disables checking for invalid characters in file names. Allows selection of file names with invalid characters.

Finally, once you get the resulting variable with value, you can do all kinds of stuff with it, such as to LAUNCH that file, store the name and exact link for document management, or use the name of that file to associate with graphics management system running R:BASE 7.0 for Windows.

Notes: Any option(s) with parameter set to OFF in examples above, may be excluded as the default is always set to OFF when not included in the syntax.

A value of '[Esc]' will be returned if clicked on [Cancel] button or Close Window [x] option of file-selection dialog.

LoadFileName.RBL plugin must be placed in C:\RBTI\RBG7 directory.

Now keeping all things described above in perspective, let's say you would like your employee(s) to pick a PDF file from any given directory and then provide the option to print that PDF file without opening the document in Adobe Acrobat Reader and then clicking on Print icon.

You can automate the entire process that will spool the PDF file to the printer without having to open Acrobat to do it. The user doesn't need to do anything to get it to print. The user will simply pick a file from the list and it will print the pdf seamlessly.

There are two methods to achieve this goal:

- ① Select the file in CHOOSE box and then send the picked file directly to the printer
- ② Use LoadFileName.RBL Plugin to navigate through the directory and then send the picked file directly to the printer

Example 01: Select the file in CHOOSE box and then send the picked file directly to the printer



```
CLEAR VAR vPDFFileName,vMsg,vPrintYesNo,vPrintString,vQuotes
SET VAR vPDFFileName TEXT = NULL
SET VAR vMsg TEXT = NULL
SET VAR vPrintYesNo TEXT = NULL
SET VAR vPrintString TEXT = NULL
SET VAR vQuotes TEXT = (CVAL('QUOTES'))

CLS
CHOOSE vPDFFileName FROM #LFILES IN *.PDF +
CHKBOX 1 TITLE 'Select PDF File to Print' +
CAPTION ' R:Pick and Print PDF' LINES 30 FORMATTED +
OPTION LIST_FONT_COLOR WHITE +
|LIST_BACK_COLOR TEAL +
|TITLE_FONT_COLOR RED +
|TITLE_BACK_COLOR WHITE +
|WINDOW_BACK_COLOR WHITE +
|TITLE_FONT_NAME VERDANA +
|TITLE_FONT_SIZE 14 +
|TITLE_BOLD ON +
|WINDOW_CAPTION +
|BUTTONS_BACK_COLOR WHITE +
|BUTTONS_SHOW_GLYPH ON

IF vPDFFileName IS NULL OR vPDFFileName = '[Esc]' THEN
    GOTO Done
ENDIF

SET VAR vMsg = +
((CHAR(013))+&'File Name:'&.vPDFFileName+(CHAR(013))+(CHAR(013))+ +
'Would you like to Print?'+(CHAR(013)))

CLS
DIALOG .vMsg vPrintYesNo vEndKey YES +
CAPTION ' R:Pick and Print PDF' +
ICON APP +
OPTION TITLE_FONT_COLOR BLACK +
|TITLE_BACK_COLOR WHITE +
|TRANSPARENCY 255 +
|WINDOW_BACK_COLOR WHITE +
|BUTTON_YES_CAPTION &Yes +
|BUTTON_NO_CAPTION &No +
|BUTTON_YES_COLOR GREEN +
|BUTTON_NO_COLOR RED +
|BUTTON_YES_FONT_COLOR WHITE +
|BUTTON_NO_FONT_COLOR WHITE
IF vPrintYesNo = 'No' OR vEndKey = '[Esc]' THEN
    GOTO Done
ENDIF
SET VAR vPrintString = +
(.vQuotes+'C:\Program Files\Adobe\Acrobat 5.0\Reader\AcroRd32.exe|/p /h' +
&.vPDFFileName+.vQuotes)
-- If using Adobe Acrobat 6.0, change the directory to 6.0 in the string above
LAUNCH &vPrintString
LABEL Done
CLEAR VAR vPDFFileName,vMsg,vPrintYesNo,vPrintString,vQuotes
RETURN
```



Example 02: Use LoadFileName.RBL Plugin to navigate through the directory and then send the picked file directly to the printer

```
CLEAR VAR vPDFFileName,vMsg,vPrintYesNo,vPrintString,vQuotes
SET VAR vPDFFileName TEXT = NULL
SET VAR vMsg TEXT = NULL
SET VAR vPrintYesNo TEXT = NULL
SET VAR vPrintString TEXT = NULL
SET VAR vQuotes TEXT = (CVAL('QUOTES'))

CLS
PLUGINS LoadFileName.RBL vPDFFileName +
|FULLPATH ON +
|TITLE R:Pick and Print PDF - Select PDF File +
|FILTER Adobe PDF Files (*.pdf)#*.PDF
IF vPDFFileName IS NULL OR vPDFFileName = '[Esc]' THEN
    GOTO Done
ENDIF

SET VAR vMsg = +
((CHAR(013))+'File Name:'&.vPDFFileName+(CHAR(013))+(CHAR(013))+ +
'Would you like to Print?'+(CHAR(013)))


CLS
DIALOG .vMsg vPrintYesNo vEndKey YES +
CAPTION ' R:Pick and Print PDF' +
ICON APP +
OPTION TITLE_FONT_COLOR BLACK +
|TITLE_BACK_COLOR WHITE +
|TRANSPARENCY 255 +
|WINDOW_BACK_COLOR WHITE +
|BUTTON_YES_CAPTION &Yes +
|BUTTON_NO_CAPTION &No +
|BUTTON_YES_COLOR GREEN +
|BUTTON_NO_COLOR RED +
|BUTTON_YES_FONT_COLOR WHITE +
|BUTTON_NO_FONT_COLOR WHITE
IF vPrintYesNo = 'No' OR vEndKey = '[Esc]' THEN
    GOTO Done
ENDIF

SET VAR vPrintString = +
(.vQuotes+'C:\Program Files\Adobe\Acrobat 5.0\Reader\AcroRd32.exe|/p /h' +
&.vPDFfilename+.vQuotes)

-- If using Adobe Acrobat 6.0, change the directory to 6.0 in the string above

LAUNCH &vPrintString

LABEL Done
CLEAR VAR vPDFFileName,vMsg,vPrintYesNo,vPrintString,vQuotes
RETURN
```



Use [RPickAndPrintPDF01.RMD](#) and [RPickAndPrintPDF02.RMD](#) command files to use Pick and Print PDFs On Demand!

Imagine the Possibilities!

Send your constructive comments to: razzak@rbase.com